EIGEN — eigenvalues and eigenvectors of a general real matrix

Purpose: EIGEN finds all the eigenvalues and eigenvectors of an N by N matrix, A.

Usage: CALL EIGEN (|NM, N, A, WR, WI, Z|)

NM → the row dimension of the two-dimensional arrays, A and Z, as specified in the dimension statements for A and Z in the calling program

N → the order of the matrix, A
N must not be greater than NM.

A → the matrix, a two-dimensional array with row dimension NM and column dimension at least N

A is overwritten during the solution.

WR ← a vector of dimension at least N, containing the real parts of the eigenvalues

WI ← a vector of dimension at least N, containing the imaginary parts of the eigenvalues

For the complex eigenvalues, the conjugate pairs are ordered so that the eigenvalue with positive imaginary part appears first in (WR, WI).

Z ← a two-dimensional array with row dimension NM, and column dimension at least N, containing the real and imaginary parts of the eigenvectors.

If the j-th eigenvalue is real, the j-th column of Z contains its eigenvector.

If the j-th eigenvalue is complex with positive imaginary part, the j-th and (j+1)-st columns of Z contain the real and imaginary parts of its eigenvector. The conjugate of this vector is the eigenvector for the conjugate eigenvalue.

Error situations: *(The user can elect to 'recover' from those errors marked with an asterisk — see *Error Handling*, Framework Chapter)

Number	Error
1	N > NM
K*	the K-th eigenvalue could not be computed after 30 iterations (the number set in [3, page 7.1-232]). The eigenvalues in the WR and WI arrays should be correct for indices, K+1, K+2 , . . . , N, but no eigenvectors have been computed.

Double-precision version: DEIGEN, with A, WR, WI, and Z declared double precision.

Storage: N real (or double-precision for DEIGEN) locations in the dynamic storage stack are used.

Method: EIGEN calls the three subroutines, ORTHE, ORTRA, and HQR2, which, in turn, are the EISPACK [2] routines, ORTHES, ORTRAN, and HQR2, (For the double-precision version, DEIGEN, the EISPACK routines have been adjusted for double precision.)

ORTHE and ORTRA transform the original matrix to upper Hessenberg form, using a sequence of orthogonal transformations. (ORTRA accumulates the transformations.) HQR2 uses the QR algorithm [1] with double origin shifts, to find the eigenvalues, and then finds the eigenvectors by back substitution.

Author: Phyllis Fox (and EISPACK)

References: [1] R. S. Martin, G. Peters and J. H. Wilkinson, The QR Algorithm for Real Hessenberg Matrices, *Numer. Math., 14*, 219-231, 1970.

[2] G. Peters and J. H. Wilkinson, Eigenvectors of Real and Complex matrices by LR and QR Triangularizations, *Numer. Math.*, *16*, 181-204, 1970.

[3] B. T. Smith, J. M. Boyle, J. J. Dongarra, B. S. Garbow, Y. Ikebe, V. C. Klema, and C. B. Moler, *Matrix Eigensystem Routines - EISPACK Guide*, Second Edition, Lecture Notes in Computer Science, Number 6, Springer - Verlag, Berlin, 1976.

[4] J. H. Wilkinson, *The Algebraic Eigenvalue Problem,* Clarendon Press, Oxford, 1965.

[5] J. H. Wilkinson and C. Reinsch, *Linear Algebra*, Handbook for Automatic Computation, Volume II, Springer - Verlag, Berlin, 1971.

Example: The program below finds the eigenvalues and eigenvectors of the following matrix (from [5] pages 394-395):

$$\begin{matrix} 3 & 1 & 2 & 5 \\ 2 & 1 & 3 & 7 \\ 3 & 1 & 2 & 4 \\ 4 & 1 & 3 & 2 \end{matrix}$$

```
      REAL A(4,4),ORT(4),Z(4,4)
      REAL H(4,4),WR(4),WI(4)
C
      DATA A(1,1),A(1,2),A(1,3),A(1,4) / 3., 1., 2., 5. /
      DATA A(2,1),A(2,2),A(2,3),A(2,4) / 2., 1., 3., 7. /
      DATA A(3,1),A(3,2),A(3,3),A(3,4) / 3., 1., 2., 4. /
      DATA A(4,1),A(4,2),A(4,3),A(4,4) / 4., 1., 3., 2. /
C
      NM=4
      N=4
C
C     SET OUTPUT WRITE UNIT
C
       IWUNIT=I1MACH(2)
C
      CALL EIGEN(NM,N,A,WR,WI,Z)
C
      WRITE (IWUNIT,96)
   96 FORMAT (22H0THE EIGENVALUES ARE -/)
C
      WRITE (IWUNIT,97) (WR(J),WI(J),J=1,N)
```

```
  97  FORMAT (/1X,2E20.8)
C
      DO 20 K=1,N
      SCALE=AMAX1(ABS(Z(1,K)),ABS(Z(2,K)),ABS(Z(3,K)),ABS(Z(4,K)))
      DO 20 J=1,N
 20   Z(J,K)=Z(J,K)/SCALE
C
      WRITE (IWUNIT,98)
  98  FORMAT (30H0THE SCALED EIGENVECTORS ARE -//)
C
      WRITE (IWUNIT,99) ((Z(J,K),K=1,N),J=1,N)
  99  FORMAT (1X,1P4E18.8/)
C
      STOP
      END
```

The results, obtained on the Honeywell 6000 computer at Bell Labs, are as follows:

```
 THE EIGENVALUES ARE -

      0.10591982E 02      0.

      0.19134652E 00      0.

    -0.23663033E 01      0.

    -0.41702519E 00      0.

 THE SCALED EIGENVECTORS ARE -

-8.57984252E-01   -2.95050379E-01    3.60162705E-01    1.12082377E-02

-1.00000000E 00    1.00000000E 00    1.00000000E 00   -1.00000000E 00

-7.83458382E-01    1.06375527E-01    9.60507886E-02    2.41136307E-01

-7.89376915E-01   -7.68113406E-02   -6.24968752E-01    9.58856940E-02
```